Is there a way to create a shortcut in Finder to an Apple Note?
I think you will like this following AppleScript solution.
STEP 1: Save this following AppleScript code as a Script File to your /Users/YOUR SHORT NAME/Library/Script Libraries/ folder. (with the name “Make Link To Note Library.scpt”).  Just compile the script and save it and don’t even bother running it from Script Editor.
This following AppleScript will be used as, what is known as a “Script Library”.
property destinationSuffix : &quot;Contents/Resources/applet.icns&quot;
property iconFile : missing value
property thisApp : missing value
property noteName : missing value

if iconFile = missing value then changeMyIconToNotesIcon()
if noteName = missing value then chooseNote()

try
    tell application &quot;Notes&quot;
        activate
        show note noteName
    end tell
end try

to changeMyIconToNotesIcon()
    set iconFile to path to resource &quot;AppIcon.icns&quot; in bundle ¬
        (path to application &quot;Notes&quot;)
    set thisApp to path to me
    set replaceThisIcon to POSIX path of thisApp &amp; ¬
        destinationSuffix as string
    do shell script &quot;cp &quot; &amp; quoted form of POSIX path of iconFile &amp; &quot; &quot; &amp; ¬
        quoted form of replaceThisIcon &amp; &quot; ; touch &quot; &amp; ¬
        quoted form of POSIX path of thisApp
end changeMyIconToNotesIcon

on chooseNote()
    tell application &quot;Notes&quot; to set noteNames to name of notes
    activate
    set noteName to (choose from list noteNames) as text
    return noteName
end chooseNote


STEP 2: Save this following AppleScript code as an application, with the name, for example “NOTES Make Link To Note Template.app”
property loadedScript : load script alias ¬
    ((path to library folder from user domain as text) &amp; ¬
        &quot;Script Libraries:Make Link To Note Library.scpt&quot;)

run loadedScript


Taking this concept one step further, and making life much easier for everyone planning to use the code in my solution. Instead of copying and pasting the code from STEP 2, into a new Script Editor document and saving it as an application etc. , every time you want to create a “Direct Link To A Specific Note” , This following STEP 3 should make the process seamless.
STEP 3: Save this following AppleScript code as an application, with the name, for example “Make Link To Note CREATOR droplet.app”
Now that we’ve created the Script Library file &quot;Make Link To Note Library. scpt”, from STEP 1’s Instructions, we can access its handlers and their commands, directly in this code by inserting the use command at the top of this following script.  This line set noteName to (notesLibrary's chooseNote()) &amp; &quot;.app” is an example of how to utilize a command from a Script Library file.
use notesLibrary : script &quot;Make Link To Note Library&quot;
use scripting additions

on open theFile
    appletFileIsAcceptable(theFile)
    
    activate
    set saveToFolder to choose folder with prompt ¬
        &quot;Save Your New \&quot;Link To Note.app\&quot; To Which Folder?&quot; multiple selections allowed false ¬
        with invisibles and showing package contents
    
    set noteName to (notesLibrary's chooseNote()) &amp; &quot;.app&quot;
    
    tell application &quot;Finder&quot;
        set duplicatedApp to duplicate item 1 of theFile to saveToFolder
        set name of duplicatedApp to noteName
        open (((saveToFolder as text) &amp; noteName) as alias)
        reveal (((saveToFolder as text) &amp; noteName) as alias)
    end tell
end open

on appletFileIsAcceptable(theFile)
    tell application &quot;System Events&quot; to if name extension of ¬
        item 1 of theFile ≠ &quot;app&quot; then return -- decide if theFile is acceptable
end appletFileIsAcceptable

In Finder, drag your new created Script Editor applet from STEP 2. directly onto the icon of the Droplet from STEP 3. This will create a duplicate of the”Template.app” to a folder of your choosing and name it to the name of the chosen Note and then launch it automatically. When the process is complete, the new “Go To Specific Note.app” will be completely configured and revealed in Finder